

Nada... ¡vale tanto! o cómo descubrir la moneda falsa sin desesperarse, ¡cualquiera que sea el número de monedas!

Rafael Losada Liste

TENEMOS 119 monedas, aparentemente iguales, de las cuales una puede pesar más o menos que las demás. Disponemos de una balanza de dos platillos. ¿Cómo podemos saber si existe una moneda «falsa» y, en este caso, cuál es, en tan sólo 5 pesadas? ¿Y si fuesen 100 monedas? ¿Y si fuesen 120 monedas? ¿Y si fuesen 37 monedas en tan sólo 4 pesadas? (*ad aeternum...*).

Introducción

Año 55 de la era espacial (2012 d.C.). Nuestro grupo participa en las IX Jornadas de Matemática Recreativa (A Coruña). Sentados en nuestras confortables butacas con pantalla de cristal líquido, nos disponemos a disfrutar de la próxima exhibición. Surge Temis, la matemaga, en el centro del escenario. Tiene algún parecido con Ada Byron. En la gran pantalla central, controlada por un ordenador HAL, aparece la figura 1.




Figura 1

En este trabajo se crea una estrategia, basada en el proceso inductivo, que permite resolver (de una vez por todas) el problema de determinar una moneda falsa, que puede pesar más o menos que el resto, entre un número N cualquiera de monedas en el mínimo número de pesadas. Este problema es la generalización del otro, más conocido, que limita N a 12.

El artículo no se limita a contemplar la posibilidad de resolución encasos concretos, sino que genera un proceso que *resuelve* de hecho todos los casos. Este proceso es de carácter algorítmico, lo que permite su informatización de forma muy sencilla. En el texto se incluye el programa informático.

**IDEAS
Y
RECURSOS**

- «Ahora necesito que alguien... Tú, por ejemplo, ¿quieres ayudarme?... Necesito que elijas una cualquiera de las 39 monedas que aparecen en la pantalla y que decidas si pesa más o menos que el resto. Por supuesto, no me digas cuál ha sido tu elección. Pero puedes escribirla en tu pantalla electrónica para que los demás la conozcan... ¿Ya está?... Bien, ahora dime qué platillos descenderían al realizar las cuatro pesadas».
- «Bajaría el platillo de la izquierda de la primera pesada, y los de la derecha de la segunda y tercera. La cuarta pesada quedaría equilibrada». (Murmullos de asentimiento.)

Casi instantáneamente la matemaga resuelve:

- «La moneda elegida fue la 25, y has decidido que pesara más».

Mi compañero de butaca sonríe:

- «Este truco ya lo conozco: basta codificar el resultado de la pesada como 0221, que no es más que el número 25 en base 3. Pero se lo voy a poner más difícil...»,

y dirigiéndose a la matemaga, añade:

- «¿Podrías volver a realizar el mismo truco con 37 monedas?».

«Tendrá que reorganizarlo todo!», pensé. Pero, para asombrar nuestra, Temis *responde*, enigmática y sonriente:

- «*Nada* me gustará más...»

y procede a realizar su truco de nuevo mientras en la pantalla grande aparece una nueva figura, similar a la anterior, pero en la que sólo figuran 37 monedas.

Cuando finaliza, oímos una voz detrás nuestra:

- «Creo que esto ya lo vi en alguna parte, hace algunos años... Me parece recordar que apareció en la revista *SUMA*, hacia finales del milenio... A propósito, ¿os habéis dado cuenta de que nuestro 55 año espacial se escribe 2001 en base 3? Y hablando de espacio, dicen que están preparando un HAL para...».

El problema

El problema consiste no tanto en resolver la cuestión para un número determinado de monedas como en hallar un procedimiento sistemático para resolver rápida y eficazmente este tipo de cuestiones. Para lograrlo, construiremos un edificio donde cada planta se eleve a partir de la anterior. Utilizando palabras más exóticas: recurriremos a un proceso inductivo.

Resolver el problema planteado, paso a paso, es un ejemplo de proceso inductivo que muestra perfectamente tanto el método como la potencia del mismo. El programa infor-

*Resolver
el problema
planteado,
paso a paso,
es un ejemplo de
proceso inductivo
que muestra
perfectamente
tanto el método
como la potencia
del mismo.
El programa
informático
que se adjunta
no es más que
el conjunto
de instrucciones
que vamos
creando
en este proceso
inductivo.*

mático que se adjunta no es más que el conjunto de instrucciones que vamos creando en este proceso inductivo. Pero este conjunto de instrucciones permite resolver de igual forma el ya clásico problema de descubrir una moneda falsa entre 12 en tres pesadas como el nuevo problema de descubrir una moneda falsa entre 3.000 en sólo ocho pesadas.

Este problema, en su versión de 12 monedas, parece ser que apareció en 1945 sin que se sepa su procedencia. Al menos así se afirma en el capítulo *El sistema ternario* (Gardner, 1986), de la sexta compilación de la famosa columna que su autor mantenía en la revista *Investigación y Ciencia* (1963). En ese capítulo, Gardner expone que una solución del problema se relaciona con los números ternarios, al igual que haremos aquí. Sin embargo, aunque expresa que *el problema ha sido generalizado*, desgraciadamente no ofrece ninguna idea que permita tal generalización y tampoco menciona cómo resolver los casos en donde haya un número de monedas que no sea de la forma PO (ver «Preparando el terreno») sin modificar drásticamente el método. Más bien parece que dicha generalización se reduce a saber cuál es el número máximo de monedas que admite cada serie de pesadas, sin profundizar en cómo se resolvería en cualquier caso. Es más, la combinación utilizada por Gardner para resolver el problema de las 12 monedas no permite, sin cambios importantes, resolver el problema con menos monedas. Digamos que es una combinación sin *nada* más. El objetivo de este artículo es cubrir estas ausencias al tiempo que se desarrolla paso a paso el proceso inductivo en uno de los problemas más atractivos de la matemática ¿recreativa?

Una sola pesada ($p=1$)

Con *una moneda*... no hay nada que hacer (incluso si es un euro, o de oro, o de chocolate). Con *dos monedas* lo más que podemos es afirmar la existencia o

no de la moneda falsa, pero no podemos determinar cuál es. Sin embargo, esta simple situación nos permite ya profundizar en el problema, al tiempo que nos apropiamos de una notación adecuada.




Figura 2

Asignamos la cifra **0** a la situación «el platillo izquierdo baja» y la cifra **2** a la situación «el platillo derecho baja». Reservamos la cifra **1** para el caso de equilibrio. Si ahora asignamos a cada moneda *el mismo número correspondiente al platillo donde se encuentra*, es evidente que sabremos inmediatamente cuál es la moneda más pesada sin más que observar cuál es el platillo que baja. Si se presenta el caso **1**, de equilibrio, al no haber ninguna moneda con ese número asignado, no existirá moneda falsa. Si se presenta otro caso, sabemos que existe una falsa, pero *no podemos saber cuál de ellas es*. Lo que ocurre es que el caso «platillo izquierdo baja» puede significar «moneda **0** pesa más» pero también «moneda **2** pesa menos».

Dos pesadas (p=2)

Ya podemos pesar 3 monedas. Hay varias formas de hacerlo. Elijamos una cualquiera, por ejemplo colocando la primera moneda sólo en el platillo derecho de la primera pesada. Esta moneda tendrá asignado el número **21** (derecho-equilibrio). Colocamos la segunda moneda en el platillo izquierdo de la primera pesada y en el derecho de la segunda. Su número asignado

será **02** (izquierdo-derecho). La tercera moneda se coloca sólo en el platillo izquierdo de la segunda pesada. Su número asignado será **10** (equilibrio-izquierdo).




Figura 3

Observemos ahora qué pasa según los casos. Si no baja ningún platillo, la moneda correspondiente será **11**, es decir, no hay moneda falsa. Si baja sólo el platillo derecho de la primera pesada, la moneda correspondiente será **21**, es decir, la primera moneda pesa más. Pero atención: si baja sólo el platillo izquierdo de la primera pesada, la moneda correspondiente será **01**, que indica «la moneda **21** pesa menos».

En la siguiente tabla se recogen todas las posibilidades:

1.ª pesada platillo que baja	2.ª pesada platillo que baja	Número correspondiente	Moneda correspondiente	¿Cómo pesa?
Izquierdo	—	01	21	Menos
Izquierdo	Derecho	02	02	Más
—	Izquierdo	10	10	Más
—	—	11	Ninguna	No hay falsa
—	Derecho	12	10	Menos
Derecho	Izquierdo	20	02	Menos
Derecho	—	21	21	Más

Tabla 1

Como vemos, simplemente anotando el resultado de cada pesada (**0**, **1**, **2**) obtenemos un número que nos indica unívocamente la moneda correspondiente: si ese número no señala directamente la moneda que pesa más, cambiamos los **0** por los **2** (y viceversa) y obtenemos la moneda que pesa menos. Dicho de otra forma, si conocemos qué número debe corresponder a cada moneda, la solución del problema es prácticamente instantánea.

Además, para establecer cómo se pesan las monedas basta saber la combinación de números que hemos asignado a cada una. En el caso anterior, basta decir que hemos pesado siguiendo la combinación (**02**, **10**, **21**).

Preparando el terreno

Adivinamos ya qué va a pasar cuando tengamos más monedas. Harán falta más pesadas, lo que significa que los números asignados a las monedas tendrán más cifras, que serán siempre **0**, **1** o **2**. Los números así expresados aparecen en el sistema ternario (base 3). Como este sistema no lo manejamos con tanta soltura como nuestro familiar sistema decimal, a partir de ahora traduciremos cada número a su correspondiente decimal. Así, la combinación anterior se convierte en {2, 3, 7}. Para conocer la posición de cada moneda, basta reconvertir estos números a la base 3.

No es posible localizar la moneda falsa entre 4 monedas en menos de 3 pesadas. Podría parecer que sí, proponiendo por ejemplo la combinación {0, 2, 6, 8}, es decir, **(00, 02, 20, 22)**. Pero recordemos que el número **22** significa tanto «moneda **22** más pesada» como «moneda **00** menos pesada». No sabríamos diferenciar la moneda **00** (desde ahora llamada *Nada*) de la **22**. Lo mismo ocurre con las monedas **02** y **20**. A estos números les llamamos *complementarios*: su suma siempre será un número en base 3 compuesto exclusivamente por **2**, lo que significa, en nuestro sistema, que su suma es siempre una potencia de 3 menos una unidad (más precisamente, dos números complementarios suman siempre $3^p - 1$, donde p es el número de pesadas). Este número es el que se conoce como *Pato* pues, en forma ternaria, sólo aparece el «pato» **2**. (Teorema 1: *Pato-Nada* son complementarios).

Otra posible ocurrencia sería la combinación {0, 4, 5, 7}, es decir, **(00, 11, 12, 21)**. Pero en este caso, si los platillos se equilibrasen con las monedas **00**, **12** y **21**, no podríamos saber si la moneda **11** es falsa o no, y en caso de serlo tampoco sabríamos si pesa más o menos que las demás. La moneda **11** no hace nada, ya que no aparece en ninguna pesada. En adelante, llamaremos siempre *Pa* al número en base 3 compuesto de exclusivamente cifras **1**. (Teorema 2: *Pa* es la mitad de *Pato*. Teorema 3: *Pa* es complementario de sí mismo.) Notación: 4^+ significará *Pa* menos 4.

Además, gracias a nuestro sistema ya podemos averiguar cuántas monedas como máximo podemos utilizar en p pesadas. Veamos: la numeración de las monedas irá de **0...0** a **2...2** lo que significa un total de 3^p números. Ahora bien, uno de ellos, *Pa*, no aparece en las pesadas. Por lo tanto tenemos $3^p - 1$ números. Cada número tiene su complementario, que señala a la misma moneda (modificando su peso). Por lo tanto, como mucho podremos disponer $(3^p - 1)/2$ monedas en los platillos. Es decir, Pa monedas. Bien, supongamos que hemos dispuesto las Pa monedas. En cada pesada aparecen 3^{p-1} monedas, repartidas en dos platillos. Pero 3^{p-1} es siempre impar, lo que impide que el número de monedas en ambos platillos sea el mismo. Siempre sobra una: por ejemplo, la moneda *Nada*. Así que

*No es posible
localizar
la moneda falsa
entre 4 monedas
en menos
de 3 pesadas.*

el número máximo de monedas que podemos distribuir con p pesadas es $Po = Pa - 1$ (Po es casi Pa). Esto quiere decir que en 2, 3, 4... pesadas podremos pesar hasta 3, 12, 39... monedas¹.

Tres pesadas ($p=3$)

Hemos visto que con 3 pesadas podemos comparar hasta 12 monedas. Hay varias formas de conseguirlo, pero estamos tratando de seguir un método, así que nos apoyaremos en las pesadas de tres monedas. Busquemos entonces cuatro combinaciones de 3 monedas que cumplan las siguientes condiciones:

- C.1) Deben indicar monedas distintas, independientemente de su peso. Debe reservarse *Nada*, para que la numeración sea de 1 a 12 (y complementarios).
- C.2) Cada combinación debe poder generar otras tres combinaciones, para que el proceso sea inductivo.
- C.3) Cada combinación debe equilibrarse por sí misma, como pasaba con la combinación {2, 3, 7}, para poder distribuir 6 o 9 monedas simplemente eliminando combinaciones sobrantes. Dicho de otra forma, en cada combinación se debe depositar una sola moneda en cada platillo.
- C.4) Podamos reemplazar dos o tres monedas por *Nada* para poder distribuir cantidades de monedas que no sean múltiplos de 3.

Sólo existen dos combinaciones que cumplan las tres primeras condiciones:

{2, 12, 25}, {6, 11, 22}, {7, 9, 23}
y {8, 10, 21} (o complementarios)

{2, 12, 25}, {6, 10, 23}, {7, 11, 21}
y {8, 9, 22} (o complementarios)

Pero la segunda no cumple la cuarta condición. Ajustamos la primera (por complementarios) para que sí la cumpla y obtenemos la *única serie posible*:

{2, 12, 25}, {4, 15, 20},
{7, 9, 23} y {8, 10, 21}

¹ Una alumna propone sustituir *Po* por el diminutivo inglés *Di* (más aristocrático), que es casi *Du* (la mitad de *Duck*).

es decir:

**{002, 110, 221}, {011, 120, 202},
{021, 100, 212}, {022, 101, 210}.**

Como *reserva*, añadimos una nueva tría-da, formada por *Nada*, *Pa* y *Pato*, que sólo utilizaremos para el proceso inductivo: {0, 13, 26} = **{000, 111, 222}**.

Veamos que cumplen todas las condiciones:

C.1) Indican monedas distintas, independientemente de su peso, pues aparecen todos los números posibles (rechazando complementarios) excepto *Nada* (000) y *Pa* (111).

C.2) Cada combinación genera 3 combinaciones nuevas sin más que antecederles **0, 1 y 2** en distinto orden. Por ejemplo, la combinación en tres pesadas {2, 12, 25}=**{002, 110, 221}** genera las combinaciones² en cuatro pesadas:

**{0 002, 1 110, 2 221},
{1 002, 2 110, 0 221},
{2 002, 0 110, 1 221},**

es decir, {2, 39, 79}, {25, 29, 66}, {12, 52, 56}. Esto hace que las 4 combinaciones se transformen en 12, equivalentes a 36 monedas. Las tres monedas que faltan hasta las 39 se consiguen recurriendo a la tríada de reserva, que genera la combinación {26, 27, 67}, su complementaria (que rechazamos) y la nueva tríada de reserva {0, 40, 80}.

C.3) Cada combinación se equilibra por sí misma, por construcción, de forma que podemos eliminar las combinaciones que queramos para pesar menos monedas. Solamente debemos tener cuidado en no eliminar las combinaciones donde aparecen 4, 10, 4' y (4'+10') si posteriormente las necesitamos.

C.4) Cuando convenga, podemos utilizar la moneda *Nada* para reemplazar a otras dos, 4 y 4':

la moneda 4' tiene asociado el número: **1...1 100**

la moneda 4 tiene asociado el número: **0...0 011**

*Y nada mejor
que un ejemplo
para ilustrar
el proceso.
Nos piden
descubrir
la moneda falsa
entre 10,
en sólo
tres pesadas.*

2 Podemos generar las combinaciones más fácilmente en nuestro sistema decimal, escribiendo tres veces la combinación original de esta forma: {2, 12, 25}, {25, 2, 12}, {12, 25, 2} y sumando $3^p = 27$ a cada segundo número y $2 \cdot 3^p = 54$ a cada tercer número. O bien, si se conoce la regla de Pierre Sarrus, escribiendo las combinaciones positivas del determinante

2	12	25
2+27	12+27	25+27
2+54	12+54	25+54

ambas se pueden sustituir por Nada: **0...0 000**

Así mismo, cuando convenga, podemos utilizar la moneda *Nada* para reemplazar a otras tres, 4, 10 y el complementario de 14:

la moneda 4 tiene asociado el número: **0...0 011**

la moneda 10 tiene asociado el número: **0...0 101**

la moneda complementaria de 14 tiene asociado el número: **2...2 110**

las tres se pueden sustituir por *Nada*: **0...0 000**

¡Ya está! Y nada mejor que un ejemplo para ilustrar el proceso. Nos piden descubrir la moneda falsa entre 10, en sólo tres pesadas. Tomamos el *pa-trón* correspondiente a 3 pesadas, que nos permite pesar hasta 12 monedas: {2, 12, 25}, {4, 15, 20}, {7, 9, 23}, {8, 10, 21}, es decir,

**{002, 110, 221}, {011, 120, 202},
{021, 100, 212}, {022, 101, 210}.**

Colocamos cada moneda en su platillo correspondiente:

Platillo Izquierdo				Platillo Derecho			
002	011	021	022	221	202	212	210
002	202	100	101	221	120	021	022
110	120	100	210	002	202	212	022

es decir:

Platillo Izquierdo				Platillo Derecho			
2	4	7	8	25	20	23	21
2	20	9	10	25	15	7	8
12	15	9	21	2	20	23	8

Con esta tabla podemos responder cuando hay doce monedas (eliminando una combinación, cuando hay 9 monedas; eliminando dos combinaciones, cuando hay 6 monedas). Como sobran dos monedas, recurrimos a *Nada*, que reemplaza a 4, 10 y 12 (complementario de 14):

Platillo Izquierdo				Platillo Derecho			
2	0	7	8	25	20	23	21
2	20	9	0	25	15	7	8
0	15	9	21	2	20	23	8

Para que se note menos lo especial de estos números, terminamos el proceso reemplazando los números mayores que *Pa* (13) por sus complementarios (*Pata* es 26):

Platillo Izquierdo				Platillo Derecho			
2	0	7	8	1	6	3	5
2	6	9	0	1	11	7	8
0	11	9	5	2	6	3	8

Hagamos la prueba. Imaginemos que la balanza queda equilibrada en la primera pesada y se inclina hacia la derecha en las demás: el número correspondiente es **122**, es decir, 17 (mayor que 13), cuyo complementario es 9: la moneda falsa es la 9. Como el 9 no aparece en los platillos hacia donde se inclina la balanza, sino en los opuestos, entonces la moneda 9 pesa menos que el resto. Evidentemente, si la balanza no se inclinara en ninguna pesada, es que no habría moneda falsa (sería la moneda *Pa*, que no interviene en las pesadas).

Por supuesto, es muy sencillo volver a diseñar la tabla para que aparezcan los números entre 1 y 10 (basta asignar a cada número uno nuevo). Sin embargo, este proceder dificulta bastante la resolución fulgurante de la matemaga.

Cientos de monedas

El procedimiento es exactamente el mismo que el descrito aquí. Los patrones para 12, 39, 120... monedas (o cualquier número intermedio) se pueden generar «a mano», mediante el proceso descrito en la condición C.2, o bien utilizar el programa informático, que nos aporta tres resultados:

- El patrón correspondiente a ese número de monedas.
- La distribución de las monedas en cada pesada.
- Realiza el truco de la matemaga.

Como ejemplo, damos a continuación el patrón de 13 tríadas (generadas por las 4 tríadas, más la tríada de reserva) que permite resolver todos los casos entre 13 y 39 monedas:

{2, 39, 79}, {25, 29, 66}, {12, 52, 56}, {4, 42, 74},
 {20, 31, 69}, {15, 47, 58}, {7, 36, 77},
 {23, 34, 63}, {9, 50, 61}, {8, 37, 75},
 {21, 35, 64}, {10, 48, 62}, {26, 27, 67};
 Reserva: {0, 40, 80}

Este patrón permite la distribución de la figura 1, presentada por HAL en la introducción. Mientras Temis procede por segunda vez a realizar el truco, lo único que tiene que hacer HAL es sustituir en la figura 1 las monedas 4, 10 y 14 por la moneda 0, lo que reduce el número de monedas a 37. Si se hubiese solicitado resolver con 28 monedas, por ejemplo, además del reemplazo anterior HAL sólo necesita eliminar 3 tríadas (en las que no aparezca el 0).

Excepción

Este método funciona para cualquier número de monedas excepto 4. El motivo de esta excepción es que no hay

*Este método
funciona para
cualquier número
de monedas
excepto 4.
El motivo
de esta excepción
es que
no hay suficientes
combinaciones
de tres monedas
para efectuar
los oportunos
reemplazos desde
el patrón de 12.
En este caso,
se emplea
cualquier
combinación
de 4 números
que se equilibren.*

suficientes combinaciones de tres monedas para efectuar los oportunos reemplazos desde el patrón de 12. En este caso, se emplea cualquier combinación de 4 números que se equilibren. En el programa informático aparece la más *elegante* de estas combinaciones.

Variante 1 del problema

Cómo descubrir la moneda falsa (sin determinar si pesa más o menos que el resto) de entre 13 (o 40,... en general, *Pa*) monedas aparentemente idénticas en tan sólo 3 pesadas:

Formamos la combinación de pesadas correspondiente a 12 de esas monedas, según las instrucciones anteriores, y procedemos de igual forma. La única variante es que sólo podemos determinar ahora si la falsa pesa más o menos en el caso de que sea una de estas 12 monedas. Si el resultado es 13, sabemos que la falsa es la moneda que no hemos pesado, pero no sabemos si pesa más o menos que el resto.

Variante 2 del problema

Cómo descubrir si existe una moneda falsa, y si pesa más o menos que el resto, de entre 13 (o 40,... en general, *Pa*) monedas aparentemente idénticas en tan solo 3 pesadas, si disponemos además de otra moneda que sabemos que no es falsa:

Formamos la combinación de pesadas correspondiente a 12 de esas monedas, según las instrucciones anteriores. Asignamos el número *Nada* a la moneda decimotercera y la añadimos en las tres pesadas. En el otro platillo, añadimos, para compensar, la moneda que sabemos no es falsa, sin asignarle número alguno. Se sigue el mismo proceso que en el problema original.

Bibliografía

GARDNER, M. (1986): *Comunicación extraterrestre y otros pasatiempos matemáticos*, Cátedra, Madrid.

Programa informático

Este programa que sigue a continuación está escrito utilizando el lenguaje QBasic, cuyo ejecutable acompaña al

Rafael Losada

IES Antonio Machado, Madrid
SMPM «Emma Castelnuevo»

sistema operativo MS-DOS (versión 6.2). He utilizado QBasic sin números de línea y sin instrucciones GOTO, es decir, empleando órdenes estructuradas, lo que facilita su lectura y la posible traducción a otro lenguaje de programación, como Visual Basic o Mathematica.

```
CLS
REM *****Solicita el número de monedas a pesar
LOCATE 10, 20: INPUT "¿Número de monedas"; B
DO WHILE B < 3: LOCATE 4, 6
PRINT "Es imposible resolver con menos de 3 monedas.": END: LOOP
DO WHILE B <> INT(B): LOCATE 4, 6
PRINT "Lo siento, pero este programa no admite decimales.": END: LOOP
SELECT CASE B
CASE 3: PRINT : PRINT "Combinación: {2, 3, 7}": PRINT
PRINT "Pesada 1"
PRINT "Platillo Izquierdo: 2   Platillo Derecho: 1"
PRINT : PRINT "Pesada 2"
PRINT "Platillo Izquierdo: 3   Platillo Derecho: 2"
PRINT : PRINT "Pato = 8, resuelve tú.": END
CASE 4: PRINT : PRINT "Combinación: {1, 4, 23, 24}": PRINT
PRINT "Pesada 1"
PRINT "Platillo Izquierdo: 1   4   Platillo Derecho: 2   3"
PRINT : PRINT "Pesada 2"
PRINT "Platillo Izquierdo: 1       Platillo Derecho: 2"
PRINT : PRINT "Pesada 3"
PRINT "Platillo Izquierdo: 2       Platillo Derecho: 3"
PRINT : PRINT "Pato = 26, resuelve tú.": END
END SELECT

REM ***Calcula los números Pa, Po y el número P de pesadas necesarias
PA = B + 1: P = LOG(2 * PA + 1) / LOG(3)
IF P <> INT(P) THEN P = INT(P) + 1
PA = (3 ^ P - 1) / 2: PO = PA - 1: F = (PA + 2) / 3

REM *****Establece el patrón inicial
DIM X(P, F + 1, 2) AS INTEGER, PI(F * 3), PD(F * 3)
DIM Y(F + 1, 2) AS INTEGER, Z(F + 1, 2) AS INTEGER
P1 = 3: PA1 = 13: X(3, 1, 0) = 2: X(3, 1, 1) = 12: X(3, 1, 2) = 25
X(3, 2, 0) = 4: X(3, 2, 1) = 15: X(3, 2, 2) = 20: X(3, 3, 0) = 7
X(3, 3, 1) = 9: X(3, 3, 2) = 23: X(3, 4, 0) = 8: X(3, 4, 1) = 10
X(3, 4, 2) = 21: X(3, 5, 0) = 0: X(3, 5, 1) = 13: X(3, 5, 2) = 26

REM Calcula iterativamente los sucesivos patrones hasta el adecuado
DO WHILE P1 < P
FOR J = 1 TO (PA1 + 2) / 3 - 1
X(P1 + 1, 3 * J - 2, 0) = X(P1, J, 0)
X(P1 + 1, 3 * J - 2, 1) = X(P1, J, 1) + 3 ^ P1
X(P1 + 1, 3 * J - 2, 2) = X(P1, J, 2) + 2 * 3 ^ P1
X(P1 + 1, 3 * J - 1, 0) = X(P1, J, 2)
X(P1 + 1, 3 * J - 1, 1) = X(P1, J, 0) + 3 ^ P1
X(P1 + 1, 3 * J - 1, 2) = X(P1, J, 1) + 2 * 3 ^ P1
X(P1 + 1, 3 * J, 0) = X(P1, J, 1)
X(P1 + 1, 3 * J, 1) = X(P1, J, 2) + 3 ^ P1
X(P1 + 1, 3 * J, 2) = X(P1, J, 0) + 2 * 3 ^ P1: NEXT
J = (PA1 + 2) / 3
X(P1 + 1, 3 * J - 2, 0) = X(P1, J, 2)
X(P1 + 1, 3 * J - 2, 1) = X(P1, J, 0) + 3 ^ P1
X(P1 + 1, 3 * J - 2, 2) = X(P1, J, 1) + 2 * 3 ^ P1
X(P1 + 1, 3 * J - 1, 0) = X(P1, J, 0)
X(P1 + 1, 3 * J - 1, 1) = X(P1, J, 1) + 3 ^ P1
X(P1 + 1, 3 * J - 1, 2) = X(P1, J, 2) + 2 * 3 ^ P1
P1 = P1 + 1: PA1 = (3 ^ P1 - 1) / 2: LOOP

REM *****Analiza si sobran combinaciones del patrón
S4 = PA - 4: C14 = 3 ^ P - 15: D = PA - 1 - B: TS = INT(D / 3)
FOR J = 1 TO F - 1: FOR K = 0 TO 2: Y(J, K) = X(P, J, K)
IF Y(J, K) = 4 THEN J4 = J: K4 = K
IF Y(J, K) = S4 THEN JS4 = J: KS4 = K
IF Y(J, K) = 10 THEN J10 = J: K10 = K
IF Y(J, K) = C14 THEN J14 = J: K14 = K
NEXT: NEXT
S = D - TS * 3
SELECT CASE S
CASE 1: Y(J4, K4) = 0: Y(JS4, KS4) = -1
```



```

CASE 2: Y(J4, K4) = 0: Y(J10, K10) = -1: Y(J14, K14) = -1
END SELECT
J = F - 1
DO WHILE FTS < TS: SIGUE = 0
    FOR K = 0 TO 2
        IF Y(J, K) < 1 THEN SIGUE = 1
    NEXT
    DO WHILE SIGUE = 0: FTS = FTS + 1
        FOR K = 0 TO 2: Y(J, K) = -1: NEXT
        SIGUE = 1: LOOP: J = J - 1: LOOP

REM *****Muestra el patrón final
CLS : LOCATE 3, 3
PRINT "El patrón para pesar las"; B; "monedas en"; P; "pesadas es:"
FOR J = 1 TO F - 1: PRINT " "; : FOR K = 0 TO 2
    IF Y(J, K) > 0 THEN Z(J, K) = 2 * PA - Y(J, K) ELSE Z(J, K) = Y(J, K)
    IF Y(J, K) > -1 THEN PRINT Y(J, K);
X(0, J, K) = Y(J, K): NEXT: NEXT
PRINT : PRINT "Pulsa cualquier tecla para continuar"
DO: LOOP UNTIL INKEY$ <> "": CLS

REM *****Muestra la disposición de las monedas
FOR I = P - 1 TO 0 STEP -1: I3 = 3 ^ I
    PRINT : PRINT "Pesada"; P - I: IP = 0: DP = 0
    FOR J = 1 TO F - 1: FOR K = 0 TO 2: CON = Y(J, K)
        DO WHILE CON > -1: M = INT(CON / I3)
            IF M = 0 THEN IP = IP + 1: PI(IP) = Z(J, K)
            CON = -1: LOOP: NEXT: NEXT
        FOR K = 2 TO IP: FOR J = 1 TO K - 1
            IF PI(J) > PI(K) THEN SWAP PI(J), PI(K)
        NEXT: NEXT
        PRINT "Platillo Izquierdo:"
        FOR K = 1 TO IP - 1: PRINT LTRIM$(STR$(PI(K))) + "-"; : NEXT
        PRINT LTRIM$(STR$(PI(IP)))
        FOR J = 1 TO F - 1: FOR K = 0 TO 2: CON = Y(J, K)
            DO WHILE CON > -1: M = INT(CON / I3)
                IF M = 2 THEN DP = DP + 1: PD(DP) = Z(J, K)
                Y(J, K) = Y(J, K) - I3 * M
                CON = -1: LOOP: NEXT: NEXT
            FOR K = 2 TO DP: FOR J = 1 TO K - 1
                IF PD(J) > PD(K) THEN SWAP PD(J), PD(K)
            NEXT: NEXT
            PRINT "Platillo Derecho:"
            FOR K = 1 TO DP - 1: PRINT LTRIM$(STR$(PD(K))) + "-"; : NEXT
            PRINT LTRIM$(STR$(PD(DP)))
            PRINT : PRINT "Pulsa cualquier tecla para continuar"
            DO: LOOP UNTIL INKEY$ <> "": NEXT

REM *****Hace el papel de matemaga (ya hizo el de HAL)
CLS : LOCATE 2, 4
PRINT "Elige una moneda y dime el resultado de cada pesada"
PRINT "Si baja el platillo izquierdo escribe I"
PRINT "Si baja el platillo derecho escribe D"
PRINT "Si no desciende ningún platillo escribe N": PRINT
FOR I = 1 TO P: A$ = ""
    DO WHILE A$ <> "I" AND A$ <> "D" AND A$ <> "N"
        LOCATE 10, 12: PRINT " "
        LOCATE 10, 12: PRINT "Pesada número" + STR$(I); : INPUT A$: LOOP
    SELECT CASE A$
        CASE "D": XX = XX + 2 * 3 ^ (P - I)
        CASE "N": XX = XX + 3 ^ (P - I)
    END SELECT: NEXT
P$ = " menos pesada ": PRINT : YY = XX
IF XX = PA THEN PRINT " No hay moneda falsa.": END
IF XX > PA THEN XX = 2 * PA - XX
SIGUE = 0
FOR J = 1 TO F - 1: FOR K = 0 TO 2
    IF XX = Z(J, K) THEN SIGUE = 1
NEXT: NEXT
IF SIGUE = 0 THEN PRINT "No es posible tal resultado!": END
FOR J = 1 TO F - 1: FOR K = 0 TO 2
    IF X(0, J, K) = YY THEN P$ = " más pesada "
NEXT: NEXT
PRINT "La moneda es la" + STR$(XX) + " y es" + P$ + "que el resto."

```